

KAERI / TR-1365 / 99



KR9900236

HBWR에서 연소 예정인 혼합 핵연료봉의 반경방향 출력분포

Radial Power Density Distribution
of MOX Fuel Rods in the HBWR

'99. 7.

한 국 원 자 력 연 구 소
Korea Atomic Energy Research Institute

31-02

①

제 출 문

한국원자력연구소장 귀하

본 보고서를 “미래형핵연료 설계기술개발” 과제와 관련하여 수행한 연구 내용의 기술보고서로 제출합니다.

제목 : HBWR에서 연소 예정인 혼합 핵연료봉의 반경방향 출력분포

1999. 7.

주저자 : 구양현 (핵연료설계기술개발팀)

공저자 : 주형국 (핵연료설계기술개발팀)

이병호 (핵연료설계기술개발팀)

손동성 (핵연료설계기술개발팀)

ABSTRACT

A subroutine FACTOR_HBWR that calculates radial power density distribution for three MOX fuel rods, which are going to be irradiated in the HBWR (Halden Boiling Water Reactor) from the beginning of the year 2000 in the framework of OECD Halden Reactor Programme (HRP), have been developed based on neutron physics results. Calculated radial power density factors by the newly developed subroutine are in good agreement with the physics calculation except slight underprediction in the central part and a little overprediction at the outer part of the pellet. The subroutine will be incorporated into a computer code COSMOS and used to analyze the in-reactor behavior of the three MOX fuel rods during the Halden irradiation test.

CONTENTS

Abstract	i
1. Introduction	1
2. Radial power depression	1
3. Results	3
4. Conclusion	3
References	4
Appendices	5
Tables	28
Figures	33

APPENDICES

Appendix 1. Derivation of constants for the program DEPRESS	5
Appendix 2. The program DEPRESS for obtaining constants a, b, c and d	7
Appendix 3. The program FACTOR_HBWR	13

TABLES

Table 1. Power density factors for MOX-1, MOX-2 and MOX-3	22
Table 2. Constants for a fitting function for power density factor	25
Table 3. Power density factors calculated by FACTOR_HBWR	26

FIGURES

Fig. 1. Radial power distribution in MOX1 at a pellet average burnup of 0 MWd/kgM	27
Fig. 2. Radial power distribution in MOX1 at a pellet average burnup of 5 MWd/kgM	28
Fig. 3. Radial power distribution in MOX1 at a pellet average burnup of 10 MWd/kgM	29
Fig. 4. Radial power distribution in MOX1 at a pellet average burnup of 20 MWd/kgM	30
Fig. 5. Radial power distribution in MOX1 at a pellet average burnup of 30 MWd/kgM	31
Fig. 6. Radial power distribution in MOX1 at a pellet average burnup of 40 MWd/kgM	32
Fig. 7. Radial power distribution in MOX1 at a pellet average burnup of 50 MWd/kgM	33
Fig. 8. Radial power distribution in MOX1 as a function of pellet average burnup	34
Fig. 9. Radial power distribution in MOX2 at a pellet average burnup of 0 MWd/kgM	35
Fig.10. Radial power distribution in MOX2 at a pellet average burnup of 5 MWd/kgM	36
Fig.11. Radial power distribution in MOX2 at a pellet average burnup of 10 MWd/kgM	37
Fig.12. Radial power distribution in MOX2 at a pellet average burnup of 20 MWd/kgM	38
Fig.13. Radial power distribution in MOX2 at a pellet average burnup of 30 MWd/kgM	39
Fig.14. Radial power distribution in MOX2 at a pellet average burnup of 40 MWd/kgM	40
Fig.15. Radial power distribution in MOX2 at a pellet average burnup of 50 MWd/kgM	41
Fig.16. Radial power distribution in MOX2 as a function of pellet average burnup	42
Fig.17. Radial power distribution in MOX3 at a pellet average burnup of 0 MWd/kgM	43
Fig.18. Radial power distribution in MOX3 at a pellet average burnup of 5 MWd/kgM	44
Fig.19. Radial power distribution in MOX3 at a pellet average burnup of 10 MWd/kgM	45
Fig.20. Radial power distribution in MOX3 at a pellet average burnup of 20 MWd/kgM	46
Fig.21. Radial power distribution in MOX3 at a pellet average burnup of 30 MWd/kgM	47
Fig.22. Radial power distribution in MOX3 at a pellet average burnup of 40 MWd/kgM	48
Fig.23. Radial power distribution in MOX3 at a pellet average burnup of 50 MWd/kgM	49
Fig.24. Radial power distribution in MOX3 as a function of pellet average burnup	50

1. Introduction

An irradiation test with three MOX fuel rods and three Inert Matrix Fuels (IMF) rods containing Pu is going to be carried out in the Halden Boiling Water Reactor (HBWR) from the beginning of the year 2000 in the framework of OECD Halden Reactor Programme (HRP). Of the three MOX fuel rods, two rods are being fabricated in the Paul Scherrer Institute (PSI), Switzerland in cooperation with the Korea Atomic Energy Research Institute (KAERI), by a dry mill method developed by the KAERI. In addition, to make a comparison with the two MOX fuel rods, one SBR MOX fuel rod manufactured and supplied by the British Nuclear Fuels Limited (BNFL) will also be irradiated together as a reference fuel. The purpose of this irradiation test is to investigate the overall in-reactor behavior of MOX and IMF fuel fabricated with new process. Main parameters measured during the test are fuel temperature and fission gas release. The detailed information about three MOX fuel rods and the outline of the irradiation test in the HBWR are given in Ref.1.

In developing and verifying a computer code that would be used for the design and performance analysis of MOX fuel, measured data from instrumented fuel assembly in the HBWR is very useful. To develop models for the analysis of MOX fuel such as fission gas release and thermal conductivity considering its microstructural characteristics dependent on fabrication method, accurate prediction of fuel temperature is a prerequisite since fuel temperature is one of the most important parameters affecting many physical processes taking place in the pellet. Therefore, the radial power depression, which governs temperature distribution across fuel pellet, should be derived to analyze the measured data effectively obtained during HBWR operation.

The purpose of the present report is, with using neutron physics calculation given in Table 1 [2], to develop a radial power distribution function corresponding to three MOX fuel rods (two rods from the KAERI/PSI and one from the BNFL).

2. Radial power distribution

In LWR fuel pellet, whether it is UO_2 or MOX pellet, radial depression of power occurs due to the self-shielding of thermal neutrons. Generally, the radial power depression in MOX pellet is different from that of conventional LWR UO_2 fuel because of the following two factors. First, neutron spectrum obtained from Pu fission is harder than that for UO_2 fuel leading to longer migration distance of thermal neutrons in the MOX pellet before they are mostly absorbed by Pu isotopes or U-238. Second, due to a rather higher total Pu content of about 6 to 8%, the amount of U-238 contained in MOX fuel is obviously less than that of typical UO_2 fuel and thus creates less Pu-239 from U-238. Furthermore, two more things in the HBWR influences the radial power depression in the MOX fuel. First, heavy water is being used as a moderator in the HBWR compared with light water in LWR resulting in different neutron spectrum in the HBWR due to less effective moderation ability of heavy water. Second, fuel-to-moderator ratio, which also determines the degree of moderation of

fission neutrons, is different from that of typical LWR. Therefore, neutron physics calculation should be made for three MOX fuel rods in the HBWR considering these four factors to obtain the radial power depression.

In general, a function $f(x)$ describing radial power distribution is derived as a function of enrichment (or Pu fissile content in case of MOX fuel), burnup and relative pellet radius based on the results of neutron physics calculations. In this calculation, quite specific radial segmentations are used. For example, in the physics calculations for the present work, a pellet was divided into seven rings of equal volume to obtain the average power density factors in each ring (see Table 1). To enable this results to be used for a general pellet division, the radial distribution of power is described as follows using a fitting function $f(x)$:

$$q'''(x) = q'''_{avg} \cdot f(x) \quad (1)$$

where

- x : relative local radius defined by r/R
- q''' : local power density
- q'''_{avg} : average power density
- f : fitting function that describes the radial power distribution

In case of LWR pellet, $f(x)$ is usually expressed by

$$f(x) = a + b \cdot x^2 + \frac{c}{(d - x^2)^n} ; n = \begin{cases} 1 & \text{for PWR} \\ 2 & \text{for BWR} \end{cases} \quad (2)$$

The constants a , b , c and d , which are functions of enrichment and pellet average burnup, are given in the Tables of Ref. [3]. Then these constants, which are tabulated in the subroutine RADPOWER of COSMOS [4], are interpolated for arbitrary enrichment, burnup and pellet radius to get $f(x)$. Finally, radial power density factor for each ring f_i is obtained by averaging the function $f(x)$ over the area between x_i and x_{i+1} .

The above function, however, was found to be inappropriate in fitting the physics calculations for MOX pellet with total Pu content of about 8% (6% of Pu fissile) in the environment of HBWR. Therefore, a new fitting function $f(x)$ with a value of 1/2 for n was tried based on the argument given below and it was found to be excellent in fitting the neutron physics calculation; that is,

$$f(x) = a + b \cdot x^2 + \frac{c}{(d - x^2)^{1/2}} \quad (3)$$

In LWR pellet, where radial power depression is described by Eq.(2), power density in pellet periphery increases with burnup due to Pu-239 formation through the absorption of epithermal neutron by U-238 while it shows opposite trend in the pellet center. On the contrary, power density in MOX pellet for the HBWR, as seen in Table 1, decreases with burnup in pellet outer area while it increases with burnup in the central part of pellet. The fitting function of Eq.(3) was selected on this basis. That is, for constants c being positive and d larger than 1

(one), which is satisfied for all cases as shown in Table 2, the last term $c / (d - x^2)^{1/2}$ in Eq.(3) simulates the situation that happens in the MOX pellet in the HBWR.

When a new fitting function is introduced, its corresponding relevant parameters such as δ , β_i , γ_i and f_i , should be derived using the procedure described in Ref.[3]. Appendix 1 shows how these parameters are derived for Eq.(3). Then the constants a , b , c and d are obtained by a computer program DEPRESS given in Appendix 2.

3. Results

Once the parameters δ , β_i , γ_i and f_i are determined, the DEPRESS is run to get the constants a , b , c and d for the density function of Eq.(3). Table 2 shows these constants for three MOX rods as a function of pellet average burnup.

A new subroutine FACTOR_HBWR given in Appendix 3 was developed to calculate the radial power density factors for three MOX fuel rods in the HBWR using a subroutine of FACTOR for LWR fuel [5]. The program was run to check the validity of the constants of Table 2. Figs.1 to 24 compare the physics results with the calculational values for 3 kinds of MOX pellets obtained by the FACTOR fitting function $f(x)$ and the constants given in Table 2. It is shown that for burnup up to about 20 MWd/kgM there is slight underprediction in the central part and a little overprediction at the outer part of the pellet. Therefore, it can be concluded that the agreement between the physics calculation and the fitting function $f(x)$ is very good. The calculated density factors are tabulated in Table 3 as a function of pellet average burnup. This subroutine will be incorporated into a computer code COSMOS and used to analyze the in-reactor behavior of the three MOX fuel rods during irradiation test.

4. Conclusion

A subroutine FACTOR_HBWR that calculates the radial power density factors for three MOX fuel rods, which are going to be fabricated by the KAERI/PSI and will be irradiated in the HBWR from the beginning of the year 2000, has been developed based on the neutron physics results. It is shown that a new fitting function of Eq.(3) reproduces well the physics calculation.

This subroutine will be incorporated into a computer code COSMOS to analyze the in-reactor behavior of the three MOX fuel rods during irradiation test.

References

- [1] Hyung-Kook Joo and Uwe Kasemeyer, "Draft: Preliminary nuclear design for IMF-MOX fuel irradiation test in the Halden Boiling Water Reactor", Dec. 1998.
- [2] Hyung-Kook Joo, Calculational file for power density of three MOX fuel rods to be irradiated in the Halden Boiling Water Reactor, June. 1999.
- [3] J. Stackmann and R. Eberle, "B111/85/100 : New best-estimate correlation for the radial power density in LWR pellet", Mar. 1985.
- [4] Yang-Hyun Koo, Byung-Ho Lee and Dong-Seong Sohn, "COSMOS: A computer code to analyze LWR UO₂ and MOX fuel up to high burnup", Annals of Nuclear Energy, Vol.26 (1999) 47-67.
- [5] Yang-Hyun Koo, "B412/90/E84 : The computer program FACTOR for calculating the radial power density and burnup factors on in fuel pellets", Mar. 1990.

Appendix 1. Derivation of constants for the program DEPRESS

The radial power density distribution described by $q'''(r) = q'''_{avg} \cdot f(r)$, should satisfy the following relation:

$$q'_{avg} = q'''_{avg} \cdot \pi R^2 = \int_0^R 2\pi r q'''(r) dr = \int_0^R 2\pi r f(r) q'''_{avg} dr,$$

where

- q'_{avg} : average linear power (W/cm)
- q'''_{avg} : average power density (W/cm³)
- q''' : local power density (W/cm³)
- R : pellet radius (cm)
- r : local pellet radius (cm) .

The above relation is reduced to

$$R^2 = 2 \int_0^R r f(r) dr.$$

Definition of x as r/R gives

$$1 = 2 \int_0^1 x f(x) dx ; f(x) = a + b \cdot x^2 + \frac{c}{(d - x^2)^{1/2}}.$$

If we let $y = x^2$, then

$$1 = \int_0^1 f(y) dy ; f(y) = a + b \cdot y + \frac{c}{(d - y)^{1/2}}.$$

Integration yields the following result:

$$a + \frac{b}{2} + c \cdot \frac{1}{2\sqrt{d}} \ln\left(\frac{\sqrt{d}+1}{\sqrt{d}-1}\right) = 1.$$

If we define δ as follows

$$\delta = \frac{1}{2\sqrt{d}} \ln\left(\frac{\sqrt{d}+1}{\sqrt{d}-1}\right),$$

the integration result is simplified to

$$a + \frac{b}{2} + c \cdot \delta = 1.$$

The average radial power density factor f_i for ring i is calculated by

$$f_i = \frac{1}{x_{i+1}^2 - x_i^2} \cdot \int_{x_i}^{x_{i+1}} f(y) dy.$$

Insertion of $f(y)$ into the above formula and integration gives

$$f_i = a + \frac{b}{2} (x_{i+1}^2 + x_i^2) + \gamma(x_i, x_{i+1}),$$

where

$$\gamma(x_i, x_{i+1}) = \frac{c}{2\sqrt{d}} \cdot \frac{1}{x_{i+1}^2 - x_i^2} \cdot \ln \frac{(\sqrt{d} + x_{i+1}^2)(\sqrt{d} - x_i^2)}{(\sqrt{d} + x_i^2)(\sqrt{d} - x_{i+1}^2)}.$$

For the division of a pellet into N equal volume, x_{i+1}^2 and x_i^2 is

$$x_i^2 = \frac{i-1}{N}$$

$$x_{i+1}^2 = \frac{i}{N}.$$

In addition, a is expressed as

$$a = 1 - \frac{b}{2} - c \cdot \delta.$$

Therefore, f_i is written as

$$f_i = 1 + \left[\frac{1}{N} \left(i - \frac{1}{2} \right) - \frac{1}{2} \right] b + \left[\frac{N}{2\sqrt{d}} \ln \frac{(\sqrt{d} + \frac{i}{N})(\sqrt{d} - \frac{i-1}{N})}{(\sqrt{d} + \frac{i-1}{N})(\sqrt{d} - \frac{i}{N})} - \delta \right] c.$$

If we define β_i and γ_i , respectively, given below

$$\beta_i = \frac{1}{N} \left(i - \frac{1}{2} \right) - \frac{1}{2}$$

$$\gamma_i = \frac{N}{2\sqrt{d}} \ln \frac{(\sqrt{d} + \frac{i}{N})(\sqrt{d} - \frac{i-1}{N})}{(\sqrt{d} + \frac{i-1}{N})(\sqrt{d} - \frac{i}{N})} - \delta,$$

f_i is finally expressed as

$$f_i = 1 + \beta_i \cdot b + \gamma_i \cdot c.$$

The program DEPRESS then can calculate the constants a , b , c and d using the parameters δ , β_i , γ_i and f_i .

Appendix 2. The program DEPRESS for obtaining constants a, b, c, and d

```
C
=====
C      PROGRAM DEPRESSION ! Depress.for
C      =====
C
C      History      : Version 0, B111/85/100 (Technical report written in German)
C      =====    : Version 1, Feb. 1996 by B.H. LEE
C                  : Version 2, June.1999 by Y.H. KOO
C
C      Purpose      : Calculates the constants of a, b, c, d for three MOX fuel rods
C      =====    : that are going to be irradiated in the HBWR from the beginning
C                  : the year of 2,000
C
C      Procedure    : [1] Obtain the data from neutron physics calculation
C      =====    : [2] Run this program
C                  : [3] Insert the constants a, b, c, d into 'FACTOR.FOR'
C                  : [4] Run the FACTOR.FOR and compare the results with the
C                  : neutron physics calculation
C
C      -----
C      DATA to be given in the INPUT File
C      -----
C 1) First Line (4 Data Points)
C      KRAD      : Reactor and Fuel Type
C                1 = PWR/UO2
C                2 = PWR/MOX
C                3 = BWR/UO2
C                4 = HBWR/MOX (KAERI/PSI Cooperation)
C      N        : # of Rings in Physics Output
C      PELRAD   : Pellet Outer Radius (mm)
C      FISS     : U-235 Enrichment of Pu-fissile Content (%)
C      BURNUP   : Burnup for Which Power Density Factors
C                (from Physics Code) are Given (MWd/kgM)
C
C 2) Second Line
C      B0       : Constant in  $f=A0+B0*(x^2)+C0/(D0-x^2)$ 
C      C0       : Constant in  $f=A0+B0*(x^2)+C0/(D0-x^2)$ 
C      D0(1)    : Constant in  $f=A0+B0*(x^2)+C0/(D0-x^2)$ 
C      DELTAD   : Constant
C
C 3) Third Line (N Data Points)
C      Radius   : Pellet Radius for Which POWER DENSITY FACTOR is Given
C                (N : # of RINGS in Physics Output)
C
C 4) Fourth Line (N Data Points)
C      G(N,K)   : POWER DENSITY FACTOR (N : # of RINGS in Physics Output)
C      -----
C      RelRing(I)=RadRing(I)/PELRAD
C      RadMidd(I)=0.5*(RadRing(I)+RadRing(I+1))
C      RelMidd(I)=RadMidd(I)/PELRAD
C      -----
C      DIMENSION G(20,20),BETA(20),GAMMA(20),V0(20,20),D0(5000),S6(5000)
C      DIMENSION RadRing(20),RadMidd(20)
```

```
DIMENSION RelRing(20),RelMid(20),RelSum(20)
DIMENSION FitPDF(20)
```

```
CHARACTER*30, FILEIN,FILEOUT
REAL IR,NR
```

```
DATA KZI/0/, KZGR/1/, F/1.E-06/, SVMIN/100./
```

```
PRINT *, " ====="
PRINT *, "      What is your input file for running CODE ?"
PRINT *, " ====="
PRINT *, " "
READ(5,*) FILEIN
PRINT *, " ====="
PRINT *, "      What is your output file after running CODE ?"
PRINT *, " ====="
PRINT *, " "
READ(5,*) FILEOUT
```

```
OPEN(UNIT=5, FILE=FILEIN)
OPEN(UNIT=6, FILE=FILEOUT)
```

```
READ(5,*) N,PELRAD,FISS,BURNUP
READ(5,*) B0,C0,D0(1),DELTAD
```

C Read-in Ring Radius Given in Physics Output: RadRing(I)

```
DO K=1,KZGR
  READ(5,*) (RadRing(I),I=1,N+1) ! Read-in RadRing (Ring Radius)
  READ(5,*) (G(I,K),I=1,N)      ! Read-in G(I,KZGR) at RadMid
END DO                          ! RadMid = 0.5 * (R(I)+R(I+1))
```

C Calculate the Relative Radii of Rings : RelRing(I)

```
DO I=1,N+1
  RelRing(I)=RadRing(I)/PELRAD
END DO
```

C Calculate the Intermediate Radius Between Rings: RadMidd(I)

C Calculate the Relative Radius Between Rings : RelMidd(I)

```
DO I=1,N
  RelSum(I) = 0.5 * ( RelRing(I)**2 + RelRing(I+1)**2 )
  RelMidd(I) = (RelSum(I)) ** 0.5
  RadMidd(I) = RelMidd(I) * PELRAD
END DO
```

```
C =====
C  ITERATION
C =====
```

```
DO 110 L=2,2000
  D0(L) = D0(1) + (L - 1) * DELTAD
```

```
C          KRAD=1 : PWR/UO2
C          KRAD=2 : PWR/MOX
C          KRAD=3 : BWR/UO2
C          KRAD=4 : HBWR/MOX
```

```

GO TO (1,2,3,4), KRAD
1 CONTINUE ! KRAD=1 : PWR/UO2
  DELTA = ALOG(1. - 1./D0(L))
  A0    = 1. - B0/2. + C0*DELTA

  DO I=1,N
    BETA(I) = (I - 0.5)/N - 0.5
    GAMMA(I)= DELTA + N*ALOG(1. + 1. / (N*D0(L) - I))
  END DO
  GO TO 90

2 CONTINUE ! KRAD=2 : PWR/MOX (=PWR/UO2)
  DELTA = ALOG(1. - 1./D0(L))
  A0    = 1. - B0/2. + C0*DELTA

  DO I=1,N
    BETA(I) = (I - 0.5)/N - 0.5
    GAMMA(I)= DELTA + N*ALOG(1. + 1. / (N*D0(L) - I))
  END DO
  GO TO 90

3 CONTINUE ! KRAD=3 : BWR/UO2
  DELTA = D0(L) * (D0(L) - 1.)
  A0    = 1. - B0/2. - C0/DELTA

  DO I=1,N
    BETA(I) = (I - 0.5)/N - 0.5
    GAMMA(I)= N**2 / ((N*D0(L)-I) * (N*D0(L)-I+1.)) - 1./DELTA
  END DO
  GO TO 90

4 CONTINUE ! KRAD=4 : HBWR/MOX

  Droot = D0(L)**0.5
  DELTA = (0.5/Droot) * ALOG((Droot + 1.) / (Droot - 1.))
  A0    = 1. - B0 /2. - C0 * DELTA

  DO I=1,N
    IR = REAL(I)
    NR = REAL(N)

    BETA(I) = (I - 0.5) / N - 0.5
    GA1 = NR / (2.*Droot)
    GA2 = (Droot + IR/NR) * (Droot - (IR-1.)/NR)
    GA3 = (Droot + (IR-1.)/NR) * (Droot - IR/NR)
    GAMMA(I)= GA1 * ALOG(GA2 / GA3) - DELTA
  END DO
  GO TO 90

90 S1 = 0.0
   S2 = 0.0
   S3 = 0.0
   S4 = 0.0
   S5 = 0.0
   S6(L) = 0.0

```



```

DO 100 K=1,KZGR
DO 100 I=1,N
V0(I,K) = G(I,K) - 1. - BETA(I) * B0 - GAMMA(I) * C0
S1 = S1 + BETA(I) * GAMMA(I)
S2 = S2 + BETA(I)**2
S3 = S3 + GAMMA(I)**2
S4 = S4 + V0(I,K) * BETA(I)
S5 = S5 + V0(I,K) * GAMMA(I)
S6(L) = S6(L) + V0(I,K)**2
100 CONTINUE

DET = S2 * S3 - S1**2
DB = (S4 * S3 - S5 * S1) / DET
DC = (S5 * S2 - S4 * S1) / DET

IF (KRAD.EQ.1) DA = DC * DELTA - DB / 2. ! PWR/UO2
IF (KRAD.EQ.2) DA = DC * DELTA - DB / 2. ! PWR/MOX
IF (KRAD.EQ.3) DA = - DC / DELTA - DB / 2. ! BWR/UO2
IF (KRAD.EQ.4) DA = - DC * DELTA - DB / 2. ! HBWR/MOX

A0 = A0 + DA
B0 = B0 + DB
C0 = C0 + DC
DAA = ABS(DA)
DBA = ABS(DB)

IF (KRAD.EQ.1) DCA = N*ABS(DC)*ALOG(1.+ 1./(N*D0(L) - N)) ! PWR/UO2
IF (KRAD.EQ.2) DCA = N*ABS(DC)*ALOG(1.+ 1./(N*D0(L) - N)) ! PWR/MOX
IF (KRAD.EQ.3) DCA = ABS(DC)/((D0(L)-1.)*(D0(L)-1.+1./N)) ! BWR/UO2
IF (KRAD.EQ.4) DCA = N*ABS(DC)*ALOG(1.+ 1./(N*D0(L)-N)) ! HBWR/MOX

KZI = KZI + 1
IF(DAA.GE.F.OR.DBA.GE.F.OR.DCA.GE.F) GO TO 90
IF(S6(L).GE.SVMIN) GO TO 110

SVMIN = S6(L)
AMIN = A0
BMIN = B0
CMIN = C0
DMIN = D0(L)

110 CONTINUE

C =====
C THE END OF ITERATION
C =====
C IF SIGMA FOR UO2 IS LESS THAN 0.0018, IT'S OK!
C IF SIGMA FOR PO2 IS LESS THAN 0.0020, IT'S OK!
C
SIGMA = SQRT(SVMIN / (N * KZGR - 3.0))

WRITE(6,901) N
WRITE(6,902) PELRAD
WRITE(6,903) FISS
WRITE(6,904) BURNUP
WRITE(6,905) KZI
901 FORMAT(' # OF RING : ', I3)

```

```

902 FORMAT(' Pellet Radius      : ', F6.2, 2x, 'mm')
903 FORMAT(' FISS Content       : ', F6.2, 2X, '%')
904 FORMAT(' Burnup             : ', F6.2, 2X, 'MWD/kGM')
905 FORMAT(' # OF ITERATION : ', I6)

WRITE(6,921)
WRITE(6,923) AMIN
WRITE(6,924) BMIN
WRITE(6,925) CMIN
WRITE(6,926) DMIN
WRITE(6,927) SIGMA
921 FORMAT(1X)
923 FORMAT(9x,' A = ', F9.5)
924 FORMAT(9x,' B = ', F9.5)
925 FORMAT(9x,' C = ', F9.5)
926 FORMAT(9x,' D = ', F9.5)
927 FORMAT(9x,' SIGMA = ', F9.5)

FitPDF0 = AMIN + BMIN*(RelRing(1)**2) +
* CMIN / (DMIN-RelRing(1)**2)
DO I=1,N
FitPDF(I)= AMIN + BMIN*(RelMid(I)**2) +
* CMIN / (DMIN-RelMid(I)**2)
END DO
FitPDFz = AMIN + BMIN*(RelRing(N+1)**2) +
* CMIN / (DMIN-RelRing(N+1)**2)

SUM = 0.
DO I=1,N
SUM = SUM + FitPDF(I)
END DO

AVG = SUM / N
FitPDF0 = FitPDF0 / AVG
DO I=1,N
FitPDF(I) = FitPDF(I) / AVG
END DO
FitPDFz = FitPDFz / AVG

WRITE (6,911)
WRITE (6,912)
WRITE (6,913)
WRITE (6,914) RadRing(1),RelRing(1),G(1,KZGR),FitPDF0
DO I=1,N
WRITE (6,915) RadRing(I),RadRing(I+1),RadMid(I),
* RelRing(I),RelRing(I+1),RelMid(I),
* G(I,KZGR), FitPDF(I)
END DO
WRITE (6,914) RadRing(N+1),RelRing(N+1),G(N,KZGR),FitPDFz
WRITE (6,916) AVG

911 FORMAT(/, ' Power Density Factor')
912 FORMAT( ' RADRI1 RADRI2 RADRID RelRI1 RelRI2 RelMid Physi
*c Fitted')
913 FORMAT( ' =====
*= =====')

```

```
914 FORMAT(1x, F6.4,36x, 3(F6.4,2x))
915 FORMAT(1x,F6.4,' - ',F6.4, 2(2x,F6.4),' - ', F6.4, 3(2x,F6.4))
916 FORMAT(//,1x,'AVG=',F6.4)
```

```
STOP
END
```

Appendix 3. The program FACTOR_HBWR

```
C =====
C PROGRAM FACTOR ! Factor_HBWR.for
C =====
C
C History      : Factor_ORI.for, Feb., 1995 by Y.H. KOO, (B412/90/E84)
C =====
C              : Factor_MOD.for, June, 1999 by Y.H. KOO
C              : (Simplified Version of Factor_ORI.for)
C              : Factor_HBWR.for, June, 1999 by Y.H. KOO
C
C Purpose      : Program to calculate radial power density factor
C =====
C
C Procedure    : [1] Receive the physics data caculated by HELIOS or FASER
C =====
C              [2] Run the Program DEPRESSION.for and get A,B,C,D as a
C                  function of enrichment (Pu-filss) and burnup
C              [3] Insert the constants A,B,C,D to this progam FACTOR.for
C                  and compare the results with those of Physics data
C              [4] Insert the constants A,B,C,D to the 'RODPOWER.for'
C
C -----
C DATA to be given in the INPUT File
C -----
C
C 1) First : KMOX,KZDIV,M,NREFBUP
C KMOX      : Key No for MOX Fuel in HBWR
C             1 = MOX1 (BNFL, Hollow, ET)
C             2 = MOX2 (KAERI, Hollow, ET)
C             3 = MOX3 (KAERI, Solid, TC)
C KZDIV     : Key No for Pellet Division
C             1 = Equal Volume
C             2 = Equal Thickness
C M         : No of Pellet Radial Rings
C Nrefbup   : No of Pellet Average Burnups at Which the Factors Are To Be
C             Calculated (Given in the Second Line, Max : 20)
C
C 2) Second : REFBUP
C REFBUP    : Pellet Average Burnups at Which the Factors Are To Be
C             Calculated (Max :20) (MWd/kgM)
C
C 3) Third : DBRI,DBRA,HBR,RDISH,VDISH,ANR
C DBRI      : Pellet Inner Diameter (mm)
C DBRA      : Pellet Outer Diameter (mm)
C HBR       : Pellet Height (mm)
C RDISH     : Dishing Radius (mm)
C VDISH     : Dishing Volume (mm3)
C ANR       : Enrichment or Pu-fiss (w/o)
C -----
C
C DIMENSION  ABFLAB(10),FLAIN(50,10),GLAIN(50,10),GBF(50,101),GLF(50,101),
*            SGBF(50,101)
C DIMENSION  R(51,20,2),HMR(50,20,2),PMR(50,20,2),ABR(50,20,2), QR(50,20)
C DIMENSION  ANR0(5,3),AB0(6,3),CA(5,6,3),CB(5,6,3),CC(5,6,3), CD(5,6,3)
```

DIMENSION A(20),B(20),C(20),D(20)
 DIMENSION AI(101), BI(101), CI(101), DI(101)
 DIMENSION REFBUP(10),PDEF(10)
 DIMENSION RRAD(51)

REAL INTPOL
 CHARACTER*30, FILEIN,FILEOUT

C

```

DATA ANR0 / 0.06 , 0. , 0. , 0. , 0. ,
*          0.06 , 0. , 0. , 0. , 0. ,
*          0.06 , 0. , 0. , 0. , 0. /
DATA AB0 / 0. , 10.E+06, 20.E+06, 30.E+06, 40.E+06, 50.E+06,
*          0. , 10.E+06, 20.E+06, 30.E+06, 40.E+06, 50.E+06,
*          0. , 10.E+06, 20.E+06, 30.E+06, 40.E+06, 50.E+06/
DATA CA/ 0.71819, 0. , 0. , 0. , 0. ,
*        0.78540, 0. , 0. , 0. , 0. ,
*        0.85077, 0. , 0. , 0. , 0. ,
*        0.92848, 0. , 0. , 0. , 0. ,
*        0.96738, 0. , 0. , 0. , 0. ,
*        0.97412, 0. , 0. , 0. , 0. ,
*        0.70715, 0. , 0. , 0. , 0. ,
*        0.78005, 0. , 0. , 0. , 0. ,
*        0.85029, 0. , 0. , 0. , 0. ,
*        0.93160, 0. , 0. , 0. , 0. ,
*        0.97123, 0. , 0. , 0. , 0. ,
*        0.97540, 0. , 0. , 0. , 0. ,
*        0.68007, 0. , 0. , 0. , 0. ,
*        0.75863, 0. , 0. , 0. , 0. ,
*        0.85883, 0. , 0. , 0. , 0. ,
*        0.92977, 0. , 0. , 0. , 0. ,
*        0.96953, 0. , 0. , 0. , 0. ,
*        0.97323, 0. , 0. , 0. , 0. /
DATA CB/ 0.33960, 0. , 0. , 0. , 0. ,
*        0.24635, 0. , 0. , 0. , 0. ,
*        0.12644, 0. , 0. , 0. , 0. ,
*        0.04024, 0. , 0. , 0. , 0. ,
*        -0.03971, 0. , 0. , 0. , 0. ,
*        -0.08057, 0. , 0. , 0. , 0. ,
*        0.34970, 0. , 0. , 0. , 0. ,
*        0.25487, 0. , 0. , 0. , 0. ,
*        0.13163, 0. , 0. , 0. , 0. ,
*        0.04069, 0. , 0. , 0. , 0. ,
*        -0.04307, 0. , 0. , 0. , 0. ,
*        -0.08305, 0. , 0. , 0. , 0. ,
*        0.43130, 0. , 0. , 0. , 0. ,
*        0.32376, 0. , 0. , 0. , 0. ,
*        0.20658, 0. , 0. , 0. , 0. ,
*        0.07449, 0. , 0. , 0. , 0. ,
*        -0.04588, 0. , 0. , 0. , 0. ,
*        -0.10957, 0. , 0. , 0. , 0. /
DATA CC/ 0.07946, 0. , 0. , 0. , 0. ,
*        0.06486, 0. , 0. , 0. , 0. ,
*        0.06102, 0. , 0. , 0. , 0. ,
*        0.02444, 0. , 0. , 0. , 0. ,

```

```

*          0.02165, 0. , 0. , 0. , 0. ,
*          0.02870, 0. , 0. , 0. , 0. ,
*          0.08371, 0. , 0. , 0. , 0. ,
*          0.06563, 0. , 0. , 0. , 0. ,
*          0.05952, 0. , 0. , 0. , 0. ,
*          0.02232, 0. , 0. , 0. , 0. ,
*          0.02031, 0. , 0. , 0. , 0. ,
*          0.02859, 0. , 0. , 0. , 0. ,
*          0.07398, 0. , 0. , 0. , 0. ,
*          0.05640, 0. , 0. , 0. , 0. ,
*          0.01772, 0. , 0. , 0. , 0. ,
*          0.01174, 0. , 0. , 0. , 0. ,
*          0.02221, 0. , 0. , 0. , 0. ,
*          0.03967, 0. , 0. , 0. , 0. /
DATA CD/ 1.19990, 0. , 0. , 0. , 0. ,
*        1.19990, 0. , 0. , 0. , 0. ,
*        1.19990, 0. , 0. , 0. , 0. ,
*        1.05470, 0. , 0. , 0. , 0. ,
*        1.02950, 0. , 0. , 0. , 0. ,
*        1.03720, 0. , 0. , 0. , 0. ,
*        1.19990, 0. , 0. , 0. , 0. ,
*        1.19990, 0. , 0. , 0. , 0. ,
*        1.19990, 0. , 0. , 0. , 0. ,
*        1.04950, 0. , 0. , 0. , 0. ,
*        1.02660, 0. , 0. , 0. , 0. ,
*        1.03650, 0. , 0. , 0. , 0. ,
*        1.19990, 0. , 0. , 0. , 0. ,
*        1.19990, 0. , 0. , 0. , 0. ,
*        1.05110, 0. , 0. , 0. , 0. ,
*        1.01390, 0. , 0. , 0. , 0. ,
*        1.03060, 0. , 0. , 0. , 0. ,
*        1.05960, 0. , 0. , 0. , 0. /

```

```

C   KMOX=1 : MOX1
C   KMOX=2 : MOX2
C   KMOX=3 : MOX3

```

```

C   FILEIN = 'Factor.in'
C   FILEOUT= 'Factor.out'

```

```

PRINT *, " ====="
PRINT *, "      What is your input file for running CODE ?"
PRINT *, " ====="
PRINT *, " "
READ(5,*) FILEIN
PRINT *, " ====="
PRINT *, "      What is your output file after running CODE ?"
PRINT *, " ====="
PRINT *, " "
READ(5,*) FILEOUT

```

```

OPEN (97, FILE = FILEIN)
OPEN (99, FILE = FILEOUT)

```

C. Read & Write Input Data

```
READ (97,*) KMOX,KZDIV,M,NREFBUP
READ (97,*) (REFBUP(I),I=1,NREFBUP)
READ (97,*) DBRI,DBRA,HBR,RDISH,VDISH,ANR
ANR = ANR / 100

WRITE (99,100)
IF (KMOX.EQ.1) WRITE (99,101)
IF (KMOX.EQ.2) WRITE (99,102)
IF (KMOX.EQ.3) WRITE (99,103)
WRITE (99,104) ANR ! MOX
100 FORMAT(/,2X,20HPELLET SPECIFICATION)
101 FORMAT ( 8X,'MOX1')
102 FORMAT ( 8X,'MOX2')
103 FORMAT ( 8X,'MOX3')
104 FORMAT ( 8X,18HPU-fiss CONTENT :,1X,2PF6.3,1X,1H%)

WRITE (99,110)
WRITE (99,111) DBRI
WRITE (99,112) DBRA
WRITE (99,113) HBR
WRITE (99,114) RDISH
WRITE (99,115) VDISH
110 FORMAT(/,2X,15HPELLET GEOMETRY)
111 FORMAT ( 8X,18HINNER DIAMETER :,1X,F6.3,1X,2Hmm)
112 FORMAT ( 8X,18HOUTER DIAMETER :,1X,F6.3,1X,2Hmm)
113 FORMAT ( 8X,18HHEIGHT :,1X,F6.3,1X,2Hmm)
114 FORMAT ( 8X,18HDISHING RADIUS :,1X,F6.3,1X,2Hmm)
115 FORMAT ( 8X,18HDISHING VOLUME :,1X,F6.3,1X,3Hmm3)

WRITE (99,116)
IF (KZDIV.EQ.1) WRITE (99,117)
IF (KZDIV.EQ.2) WRITE (99,118)
WRITE (99,119) M
116 FORMAT (/,2X,15HPELLET DIVISION)
117 FORMAT ( 8X,12HEQUAL VOLUME)
118 FORMAT ( 8X,15HEQUAL THICKNESS)
119 FORMAT ( 8X,18HNo of RAD. RINGS :,1X,I2)
```

C.(0) CALCULATE THE RADIUS OF EACH ANNULUS

C

```
CALL GEOKALT (M,KZDIV)
```

```
L = KMOX
```

```
NABFLAB = 6
```

C.(1) INTERPOLATE THE CONSTANTS FOR THE GIVEN SPECIFIC ENRICHMENT

C (ANR)

```
DO I=1,NABFLAB
  A(I) = CA(1,I,L)
  B(I) = CB(1,I,L)
  C(I) = CC(1,I,L)
  D(I) = CD(1,I,L)
END DO
```

C.(2) CALCULATE THE GBF(J,I) FOR EACH BURNUP AND EACH ANNULUS UP TO
 C BURNUP AB0(NABFLAB,L)

IBUPEND = IFIX (AB0(NABFLAB,L)/1.E+06)
 IBUPEND = IBUPEND + 1

DO 8 I=1,IBUPEND !!! Outer DO LOOP
 BUP=(I-1) * 1.E+06
 DO II=2,NABFLAB
 IB=II
 IF (BUP.LE.AB0(II,L)) GO TO 10
 END DO
 IB=NABFLAB
 10 CONTINUE

AI(I) = INTPOL(A(IB-1),A(IB),AB0(IB-1,L),AB0(IB,L),BUP)
 BI(I) = INTPOL(B(IB-1),B(IB),AB0(IB-1,L),AB0(IB,L),BUP)
 CI(I) = INTPOL(C(IB-1),C(IB),AB0(IB-1,L),AB0(IB,L),BUP)
 DI(I) = INTPOL(D(IB-1),D(IB),AB0(IB-1,L),AB0(IB,L),BUP)

DO 12 J=1,M !!! Inner DO LOOP
 RRQA = R(J+1,1,2) / R(M+1,1,2)
 RRQI = R(J,1,2) / R(M+1,1,2)
 RQA = RRQA ** 2
 RQI = RRQI ** 2
 RQM = (RQI + RQA) * 0.5
 GO TO (13,13,13), KMOX

13 GAMMA = 0 ! MOX1, MOX2, MOX3
 G1 = CI(I) / (2.*DI(I)**0.5)
 G2 = 1. / (RQA - RQI)
 G3 = (DI(I)**0.5 + RQA) * (DI(I)**0.5 - RQI)
 G4 = (DI(I)**0.5 + RQI) * (DI(I)**0.5 - RQA)
 G5 = ALOG(G3/G4)
 GAMMA = G1 * G2 * G5
 GBF(J,I) = AI(I) + BI(I)*RQM + GAMMA
 GO TO 12

12 CONTINUE !!! Inner DO LOOP
 8 CONTINUE !!! Outer DO LOOP

C.(3) CALCULATE THE RELATIVE RADIUS OF EACH ANNULUS

RRAD(1) = DBRI / DBRA
 DO J = 1, M
 RRAD(J+1) = R(J+1,1,2) / R(M+1,1,2)
 END DO

C.(4) CALCULATE THE POWER DENSITY FACTOR GBF(J,I) FOR EACH ANNULUS AT
 C EACH INTEGER BURNUP

DO 16 I = 1,IBUPEND
 GBFSUM = 0.
 DO J=1,M
 GBFSUM = GBFSUM + GBF(J,I)
 END DO


```

    GBFKOR = M / GBFSUM ! NORMALIZING FACTOR
    DO J=1,M
      GBF(J,I) = GBF(J,I) * GBFKOR
    END DO
16 CONTINUE

```

C.(5) PROCEDURE BELOW IS NECESSARY TO CALCULATE THE RING BURNUP
 C FACTOR FOR BURNUPS GREATER THAN $AB0(NABFLAB,L)$, THE LARGEST
 C BURNUP FOR WHICH THE CONSTANTS CA, CB, CC, AND CD ARE GIVEN

```

    IBUPEN1 = IBUPEND + 1
    DO I = IBUPEN1, 101
      DO J = 1, M
        GBF(J,I) = GBF(J,IBUPEND)
      END DO
    END DO

```

C.(6) CALCULATE THE BURNUP FACTORS FOR EACH ANNULUS AT EACH
 C INTEGER BURNUP

```

    DO 20 I = 1,101
      IF (I.EQ.1) GO TO 26

      DO J = 1,M
        SGBF(J,I) = (GBF(J,1)+GBF(J,I)) * 0.5
        IF (I.EQ.2) GO TO 25
        DO K = 2,I-1
          SGBF(J,I) = SGBF(J,I) + GBF(J,K)
        END DO
25      GLF(J,I) = SGBF(J,I) / (I-1)
      END DO
      GLFSUM = 0
      DO J = 1,M
        GLFSUM = GLFSUM + GLF(J,I)
      END DO
      GLFKOR = M / GLFSUM

      DO J = 1,M
        GLF(J,I) = GLF(J,I) * GLFKOR
      END DO
      GO TO 20
26 CONTINUE

      DO J = 1,M
        GLF(J,I) = GBF(J,I)
      END DO
20 CONTINUE

```

C.(7) CALCULATE THE "NORMALIZED" POWER DENSITY FACTOR FLAIN(J,I)
 C & BURNUP FACTOR GLAIN(J,I) FOR THE GIVEN BURNUP REFBUP(I)

```

    DO 41 I = 1,NREFBUP
      BP = REFBUP(I)
      DO II = 1,101
        IN = II
        IF (BP.LE.II) GO TO 43
      END DO

```

```

      IN=101
43  CONTINUE

      DO J = 1,M
          R1 = IN-1
          R2 = IN
          FLAIN(J,I) = INTPOL(GBF(J,IN),GBF(J,IN+1),R1,R2,BP)
          GLAIN(J,I) = INTPOL(GLF(J,IN),GLF(J,IN+1),R1,R2,BP)
      END DO
41  CONTINUE

      DO 51 I = 1,NREFBUP
          FLASUM = 0.0
          GLASUM = 0.0
          DO J = 1,M
              FLASUM = FLASUM + (RRAD(J+1)**2 - RRAD(J)**2) * FLAIN(J,I)
              GLASUM = GLASUM + (RRAD(J+1)**2 - RRAD(J)**2) * GLAIN(J,I)
          END DO

          FKOR = (1 - RRAD(1)**2) / FLASUM
          GKOR = (1 - RRAD(1)**2) / GLASUM

          DO J = 1,M
              FLAIN(J,I) = FLAIN(J,I) * FKOR
              GLAIN(J,I) = GLAIN(J,I) * GKOR
          END DO
51  CONTINUE

      DO 54 I = 1,NREFBUP
          FLASUM = 0.0
          GLASUM = 0.0
          DO J = 1,M
              FLASUM = FLASUM + FLAIN(J,I)
              GLASUM = GLASUM + GLAIN(J,I)
          END DO

          FKOR = M / FLASUM
          GKOR = M / GLASUM

          DO J = 1,M
              FLAIN(J,I) = FLAIN(J,I) * FKOR
              GLAIN(J,I) = GLAIN(J,I) * GKOR
          END DO
54  CONTINUE

C.(9) PRINT OUT POWER DENSITY FACTORS AND BURNUP FACTORS
C...PRINT OUT POWER DENSITY FACTORS, FLAIN(J,I)

      WRITE (99,1000)
      WRITE (99,1001) (REFBUP(I), I=1,NREFBUP)
      WRITE (99,1002)
      WRITE (99,1003)

      DO J = 1,M
          WRITE (99,1004) J, RRAD(J+1), (FLAIN(J,I),I=1,NREFBUP)
      END DO

```

```

1000 FORMAT (///,2X,22H1.POWER DENSITY FACTOR)
1001 FORMAT ( /,4X,15HBURNUP(MWD/KGU), 7X,10(OPF6.2,3X))
1002 FORMAT ( 4X,7HNODE NO,5X,6HRADIUS)
1003 FORMAT ( 4X,7H=====,5X,6H=====)
1004 FORMAT ( 7X,I2,7X,0PF6.4, 10(4X,10(OPF6.4,3X)))

```

```

STOP
END

```

```

REAL FUNCTION INTPOL(Y1,Y2,X1,X2,X)
INTPOL = Y1 + (X-X1)/(X2-X1) * (Y2-Y1)
RETURN
END

```

```

SUBROUTINE GEOKALT (M,KZDIV)

```

```

C
C
C

```

```

ANNULAR PARTITION OF PELLETT AND COLD GEOMETRY

```

```

DIMENSION R(51,20,2),HMR(50,20,2),PMR(50,20,2),ABR(50,20,2), QR(50,20)
DIMENSION RREL2(51),RK(51),HMRK(50)

```

```

REAL MBRGES,MBRPEL,MBRRING
DATA PI /3.14159265359/

```

```

RZK = 0.5*DBRI
GO TO (1,9), KZDIV

```

```

1 IF (VDISH.NE.0.0) GO TO 2
IF (VDISH.EQ.0.0) GO TO 3

```

```

2 VPELLET=PI/4*DBRA**2*HBR*(1-(DBRI/DBRA)**2)-
* VDISH*(1-(DBRI/2./RDISH)**2)**2
GO TO 4

```

```

3 VPELLET=PI/4*DBRA**2*HBR*(1-(DBRI/DBRA)**2)

```

```

4 CONTINUE
VRING0 = VPELLET / M
VREL = VRING0 / ( PI * HBR * DBRA**2 ) * 4.
RDREL2 = (2.*RDISH/DBRA)**2
RZREL2 = (DBRI/DBRA)**2

```

```

IF (RDREL2.GT.1..OR.RZREL2.GE.1.) RETURN
HDREL = 0.
IF (RDISH.GT.RZK) HDREL = VDISH/(PI*HBR*RDISH**2)

```

```

RREL2(1) = RZREL2
IF (HDREL.NE.0.) GO TO 5
JMIN = 1
GO TO 7

```

```

5 CONTINUE
DO J=1,M
X = RDREL2*(1.-2.*HDREL*(1.-RREL2(J)/RDREL2))/(2.*HDREL)
DELTAR2 = SQRT(RDREL2*VREL/HDREL+X*X) - X
RREL2(J+1) = RREL2(J) + DELTAR2
IF (RREL2(J+1).GT.RDREL2) GO TO 6

```

```

END DO
RETURN

6 CONTINUE
RREL2(J+1) = RREL2(J)+VREL+HDREL*RDREL2*(1.-RREL2(J)/RDREL2)**2
NRDISHR    = J
IF (J.EQ.M) GO TO 8
JMIN = J + 1

7 CONTINUE
NRDISHR = JMIN
DO J=JMIN,M
  RREL2(J+1) = RREL2(J) + VREL
END DO

8 CONTINUE
RK(1) = RZK
DO J=1,M
  RK(J+1) = 0.5 * DBRA * SQRT(RREL2(J+1))
  HMRK(J) = HBR * VREL / ( RREL2(J+1) - RREL2(J) )
END DO

R(1,1,2)= RK(1)
DO J=1,M
  R (J+1,1,2)= RK(J+1)
  PMR (J,1,2) = PORNEU
  HMR (J,1,2)= HMRK(J)
  ABR (J,1,2)= 0.
END DO
RETURN

9 THICK = (DBRA - DBRI)/2/M
DO J = 1,M+1
  R(J,1,2)=(J-1) * THICK + RZK
END DO

RETURN
END

```

Table 1a. Power density factors for MOX-1 [2]

	MOX-1		Outer Ring Radius (mm)								
Burnup(MWD/kg)	0.9	1.58004	2.2363	2.7379	3.16008	3.5321	3.87068	4.18			
1	0.0000	0.0000	0.8188	0.8411	0.8917	0.9563	1.0312	1.1332	1.2696		
2	0.0000	0.0000	0.8147	0.8375	0.8891	0.9552	1.0316	1.1362	1.2762		
3	0.0500	0.0000	0.8148	0.8377	0.8892	0.9552	1.0316	1.1361	1.2759		
4	0.5000	0.0000	0.8164	0.8391	0.8903	0.9558	1.0314	1.1347	1.2734		
5	1.0000	0.0000	0.8183	0.8408	0.8916	0.9564	1.0311	1.1329	1.2704		
6	2.0000	0.0000	0.8230	0.8451	0.8948	0.9579	1.0304	1.1288	1.2632		
7	3.0000	0.0000	0.8278	0.8494	0.8980	0.9593	1.0296	1.1246	1.2561		
8	4.0000	0.0000	0.8327	0.8538	0.9012	0.9607	1.0287	1.1203	1.2489		
9	5.0000	0.0000	0.8376	0.8582	0.9044	0.9621	1.0277	1.1160	1.2420		
10	6.0000	0.0000	0.8425	0.8626	0.9075	0.9634	1.0267	1.1116	1.2351		
11	7.0000	0.0000	0.8475	0.8670	0.9106	0.9647	1.0256	1.1073	1.2284		
12	8.0000	0.0000	0.8524	0.8713	0.9137	0.9659	1.0245	1.1029	1.2218		
13	9.0000	0.0000	0.8574	0.8757	0.9168	0.9671	1.0233	1.0986	1.2154		
14	10.0000	0.0000	0.8623	0.8801	0.9197	0.9682	1.0221	1.0943	1.2091		
15	15.0000	0.0000	0.8868	0.9014	0.9340	0.9729	1.0154	1.0727	1.1804		
16	20.0000	0.0000	0.9107	0.9220	0.9469	0.9761	1.0076	1.0520	1.1561		
17	25.0000	0.0000	0.9333	0.9409	0.9581	0.9777	0.9993	1.0325	1.1369		
18	30.0000	0.0000	0.9536	0.9577	0.9671	0.9779	0.9907	1.0153	1.1231		
19	35.0000	0.0000	0.9707	0.9714	0.9737	0.9767	0.9824	1.0009	1.1150		
20	40.0000	0.0000	0.9837	0.9812	0.9777	0.9745	0.9751	0.9901	1.1126		
21	45.0000	0.0000	0.9914	0.9870	0.9790	0.9718	0.9697	0.9837	1.1149		
22	50.0000	0.0000	0.9936	0.9882	0.9784	0.9694	0.9666	0.9818	1.1202		

Table 1b. Power density factors for MOX-2 [2]

	MOX-2		Outer Ring Radius (mm)						
Burnup(MWD/kg)	0.9	1.58004	2.2363	2.7379	3.16008	3.5321	3.87068	4.18	
1	0.0000	0.0000	0.8122	0.8354	0.8876	0.9541	1.0317	1.1379	1.2808
2	0.0000	0.0000	0.8092	0.8328	0.8857	0.9531	1.0321	1.1400	1.2858
3	0.0500	0.0000	0.8094	0.8330	0.8858	0.9532	1.0320	1.1399	1.2854
4	0.5000	0.0000	0.8109	0.8343	0.8869	0.9537	1.0319	1.1386	1.2830
5	1.0000	0.0000	0.8128	0.8361	0.8882	0.9545	1.0316	1.1368	1.2798
6	2.0000	0.0000	0.8176	0.8405	0.8915	0.9561	1.0309	1.1326	1.2722
7	3.0000	0.0000	0.8225	0.8450	0.8948	0.9577	1.0301	1.1284	1.2645
8	4.0000	0.0000	0.8276	0.8495	0.8981	0.9592	1.0293	1.1240	1.2569
9	5.0000	0.0000	0.8326	0.8540	0.9014	0.9607	1.0285	1.1196	1.2494
10	6.0000	0.0000	0.8377	0.8586	0.9047	0.9621	1.0275	1.1152	1.2420
11	7.0000	0.0000	0.8429	0.8632	0.9081	0.9635	1.0265	1.1107	1.2347
12	8.0000	0.0000	0.8480	0.8677	0.9113	0.9649	1.0254	1.1062	1.2276
13	9.0000	0.0000	0.8532	0.8723	0.9146	0.9662	1.0242	1.1017	1.2206
14	10.0000	0.0000	0.8584	0.8769	0.9178	0.9674	1.0231	1.0973	1.2138
15	15.0000	0.0000	0.8841	0.8995	0.9330	0.9728	1.0164	1.0749	1.1823
16	20.0000	0.0000	0.9094	0.9212	0.9468	0.9764	1.0085	1.0531	1.1556
17	25.0000	0.0000	0.9333	0.9412	0.9588	0.9784	0.9999	1.0326	1.1345
18	30.0000	0.0000	0.9547	0.9590	0.9684	0.9787	0.9909	1.0145	1.1195
19	35.0000	0.0000	0.9727	0.9733	0.9753	0.9774	0.9823	0.9994	1.1110
20	40.0000	0.0000	0.9859	0.9834	0.9792	0.9750	0.9747	0.9885	1.1089
21	45.0000	0.0000	0.9933	0.9887	0.9801	0.9719	0.9692	0.9823	1.1126
22	50.0000	0.0000	0.9945	0.9890	0.9786	0.9691	0.9663	0.9809	1.1200

Table 1c. Power density factors for MOX-3 [2]

	MOX-3			Outer Ring Radius (mm)					
Burnup(MWD/kg)	1.58004	2.2363	2.7379	3.16008	3.5321	3.87068	4.18		
1	0.0000	0.7827	0.8310	0.8984	0.9638	1.0586	1.1556	1.3107	
2	0.0000	0.7781	0.8272	0.8959	0.9627	1.0596	1.1590	1.3183	
3	0.0500	0.7783	0.8274	0.8960	0.9627	1.0596	1.1589	1.3179	
4	0.5000	0.7800	0.8289	0.8971	0.9633	1.0592	1.1574	1.3150	
5	1.0000	0.7821	0.8308	0.8984	0.9639	1.0587	1.1554	1.3114	
6	2.0000	0.7874	0.8353	0.9016	0.9655	1.0574	1.1508	1.3027	
7	3.0000	0.7929	0.8399	0.9048	0.9670	1.0560	1.1460	1.2941	
8	4.0000	0.7984	0.8446	0.9080	0.9684	1.0545	1.1412	1.2855	
9	5.0000	0.8041	0.8493	0.9112	0.9698	1.0530	1.1364	1.2769	
10	6.0000	0.8098	0.8540	0.9144	0.9711	1.0513	1.1315	1.2685	
11	7.0000	0.8156	0.8588	0.9175	0.9724	1.0496	1.1265	1.2603	
12	8.0000	0.8214	0.8635	0.9206	0.9735	1.0478	1.1216	1.2521	
13	9.0000	0.8272	0.8683	0.9236	0.9747	1.0459	1.1166	1.2441	
14	10.0000	0.8331	0.8730	0.9267	0.9758	1.0440	1.1116	1.2364	
15	15.0000	0.8625	0.8964	0.9407	0.9802	1.0335	1.0867	1.2006	
16	20.0000	0.8918	0.9188	0.9532	0.9826	1.0217	1.0626	1.1699	
17	25.0000	0.9200	0.9395	0.9634	0.9832	1.0091	1.0397	1.1454	
18	30.0000	0.9461	0.9579	0.9713	0.9822	0.9963	1.0193	1.1274	
19	35.0000	0.9688	0.9728	0.9764	0.9796	0.9842	1.0024	1.1162	
20	40.0000	0.9864	0.9836	0.9788	0.9760	0.9738	0.9897	1.1119	
21	45.0000	0.9976	0.9896	0.9786	0.9721	0.9663	0.9822	1.1137	
22	50.0000	1.0017	0.9907	0.9766	0.9689	0.9625	0.9800	1.1199	

Table 2. Constants for a fitting function for power density factors

Burnup Rod	Constant	0.0	5.0	10.0	20.0	30.0	40.0	50.0
MOX 1 (Hollow, ET ¹)	a	0.71819	0.74831	0.78540	0.85077	0.92848	0.96738	0.97412
	b	0.33960	0.30063	0.24635	0.12644	0.04024	-0.03971	-0.08057
	c	0.07946	0.07192	0.06486	0.06102	0.02444	0.02165	0.02870
	d	1.19990	1.19990	1.19990	1.19990	1.05470	1.02950	1.03720
MOX 2 (Hollow, ET ¹)	a	0.70715	0.74059	0.78005	0.85029	0.93160	0.97123	0.97540
	b	0.34970	0.30963	0.25487	0.13163	0.04069	-0.04307	-0.08305
	c	0.08371	0.07420	0.06563	0.05952	0.02232	0.02031	0.02859
	d	1.19990	1.19990	1.19990	1.19990	1.04950	1.02660	1.03650
MOX 3 (Solid, TC ²)	a	0.68007	0.71506	0.75863	0.85883	0.92977	0.96953	0.97323
	b	0.43130	0.38758	0.32376	0.20658	0.07449	-0.04588	-0.10957
	c	0.07398	0.06467	0.05640	0.01772	0.01174	0.02221	0.03967
	d	1.19990	1.19990	1.19990	1.05110	1.01390	1.03060	1.05960

1 : Extension Thermometer

2 : Thermocouple

Table 3. Power density factors calculated by FACTOR_HBWR

1) MOX 1

Burnup (MWd/kgM)	<u>0.00</u>	<u>5.00</u>	<u>10.00</u>	<u>20.00</u>	<u>30.00</u>	<u>40.00</u>	<u>50.00</u>
Relative Radius							
0.4324	0.8154	0.8383	0.8614	0.9120	0.9540	0.9831	0.9925
0.5697	0.8647	0.8811	0.8976	0.9317	0.9608	0.9788	0.9829
0.6791	0.9167	0.9263	0.9359	0.9536	0.9688	0.9757	0.9749
0.7725	0.9727	0.9752	0.9776	0.9787	0.9787	0.9745	0.9694
0.8551	1.0365	1.0311	1.0257	1.0100	0.9926	0.9773	0.9690
0.9304	1.1193	1.1043	1.0893	1.0559	1.0179	0.9917	0.9835
1.0000	1.2747	1.2437	1.2125	1.1581	1.1273	1.1190	1.1278

2) MOX 2

Burnup (MWd/kgM)	<u>0.00</u>	<u>5.00</u>	<u>10.00</u>	<u>20.00</u>	<u>30.00</u>	<u>40.00</u>	<u>50.00</u>
Relative Radius							
0.4324	0.8088	0.8331	0.8575	0.9108	0.9553	0.9855	0.9934
0.5697	0.8597	0.8773	0.8949	0.9312	0.9620	0.9806	0.9834
0.6791	0.9133	0.9239	0.9344	0.9536	0.9698	0.9769	0.9751
0.7725	0.9713	0.9743	0.9774	0.9792	0.9795	0.9750	0.9692
0.8551	1.0374	1.0321	1.0268	1.0108	0.9928	0.9769	0.9685
0.9304	1.1234	1.1077	1.0919	1.0568	1.0169	0.9899	0.9827
1.0000	1.2860	1.2517	1.2172	1.1576	1.1236	1.1153	1.1276

3) MOX 3

Burnup (MWd/kgM)	<u>0.00</u>	<u>5.00</u>	<u>10.00</u>	<u>20.00</u>	<u>30.00</u>	<u>40.00</u>	<u>50.00</u>
Relative Radius							
0.3820	0.7720	0.8001	0.8283	0.8900	0.9463	0.9876	1.0027
0.5391	0.8368	0.8569	0.8770	0.9207	0.9576	0.9819	0.9884
0.6589	0.9039	0.9157	0.9274	0.9521	0.9695	0.9774	0.9762
0.7594	0.9747	0.9777	0.9807	0.9846	0.9824	0.9749	0.9672
0.8473	1.0525	1.0459	1.0393	1.0198	0.9974	0.9762	0.9646
0.9268	1.1478	1.1295	1.1113	1.0631	1.0188	0.9890	0.9796
1.0000	1.3123	1.2742	1.2361	1.1697	1.1279	1.1130	1.1212




Fig.1. Radial power distribution in MOX1 at a pellet average burnup of 0 MWd/kgM




Fig.2. Radial power distribution in MOX1 at a pellet average burnup of 5 MWd/kgM




Fig.3. Radial power distribution in MOX1 at a pellet average burnup of 10 MWd/kgM




Fig.4. Radial power distribution in MOX1 at a pellet average burnup of 20 MWd/kgM




Fig.5. Radial power distribution in MOX1 at a pellet average burnup of 30 MWd/kgM




Fig.6. Radial power distribution in MOX1 at a pellet average burnup of 40 MWd/kgM




Fig.7. Radial power distribution in MOX1 at a pellet average burnup of 50 MWd/kgM




Fig.8. Radial power distribution in MOX1 as a function of pellet average burnup




Fig.9. Radial power distribution in MOX2 at a pellet average burnup of 0 MWd/kgM




Fig.10. Radial power distribution in MOX2 at a pellet average burnup of 5 MWd/kgM




Fig.11. Radial power distribution in MOX2 at a pellet average burnup of 10 MWd/kgM




Fig.12. Radial power distribution in MOX2 at a pellet average burnup of 20 MWd/kgM




Fig.13. Radial power distribution in MOX2 at a pellet average burnup of 30 MWd/kgM




Fig.14. Radial power distribution in MOX2 at a pellet average burnup of 40 MWd/kgM




Fig.15. Radial power distribution in MOX2 at a pellet average burnup of 50 MWd/kgM




Fig.16. Radial power distribution in MOX2 as a function of pellet average burnup




Fig.17. Radial power distribution in MOX3 at a pellet average burnup of 0 MWd/kgM




Fig.18. Radial power distribution in MOX3 at a pellet average burnup of 5 MWd/kgM




Fig.19. Radial power distribution in MOX3 at a pellet average burnup of 10 MWd/kgM




Fig.20. Radial power distribution in MOX3 at a pellet average burnup of 20 MWd/kgM




Fig.21. Radial power distribution in MOX3 at a pellet average burnup of 30 MWd/kgM




Fig.22. Radial power distribution in MOX3 at a pellet average burnup of 40 MWd/kgM




Fig.23. Radial power distribution in MOX3 at a pellet average burnup of 50 MWd/kgM




Fig.24. Radial power distribution in MOX3 as a function of pellet average burnup

서지정보양식

수행기관보고서번호	위탁기관보고서번호	표준보고서번호	INIS 주제코드
KAERI/TR-1365/99			
제목/부제			
HBWR에서 연소 예정인 혼합 핵연료봉의 반경방향 출력분포			
연구책임자 및 부서명 (TR 경우 주저자)	구양현 (핵연료설계기술개발팀)		
연구자 및 부서명	주형국 / 이병호 / 손동성 (핵연료설계기술개발팀)		
발행지	대전	발행기관	한국원자력연구소
발행연도	1999. 7.		
페이지	P. 50	도표	있음(o) 없음()
크기	19 × 26 cm		
참고사항			
비밀여부	공개(o), 대외비(), ___ 비밀	보고서 종류	기술보고서
연구위탁기관			계약번호
초록 (15-20줄 내외)			
<p>KAERI가 스위스의 PSI와 협력하여 공동으로 제조하는 두 개의 혼합 핵연료봉과 BNFL이 제공하는 한 개의 기준 혼합 핵연료봉이 함께, KAERI가 개발한 새로운 공정으로 제조된 혼합 핵연료봉의 노내 거동을 파악하기 위해, OECD HRP의 일환으로 2000년 초반부터 HBWR에 장전되어 연소될 예정이다.</p> <p>HBWR에 장전되어 연소될 때 이 세 개의 핵연료봉의 노내 거동을 예측하고 분석하기 위해서는 우선적으로 핵연료의 온도를 정확하게 계산할 수 있어야 한다. 따라서 핵연료 온도에 영향을 미치는 반경방향 출력분포를 고려해야 한다. 세 개의 혼합 핵연료봉에 대한 핵 계산 결과를 바탕으로 HBWR에서의 혼합 핵연료봉에 대한 반경방향 출력분포를 계산하는 부 프로그램인 FACTOR_HBWR를 개발하였다. FACTOR_HBWR의 계산 값을 핵 계산 값과 비교한 결과 소결체 중심부에서 약간 작게 계산하는 것과 소결체 외각에서 약간 크게 계산하는 것을 제외하고는 FACTOR_HBWR은 핵 계산 값을 잘 재현하고 있다. 이 부 프로그램은 COSMOS에 삽입되어 HBWR에서의 혼합 핵연료봉 노내거동 분석에 활용될 계획이다.</p>			
주제명키워드	반경방향 출력 분포, 출력 분포 인자, 혼합 핵연료, HBWR, HRP, KAERI, PSI, FACTOR, COSMOS		

BIBLIOGRAPHIC INFORMATION SHEET							
Performing Org. Report No.		Sponsoring Org. Report No.		Standard Report No.		INIS Subject Code	
KAERI/TR-1365/99							
Title / Subtitle		Radial Power Density Distribution of MOX Fuel Rods in the HBWR					
Main Author		Yang-Hyun KOO (Future Fuel Development Division)					
Co-author		Hyung-Kook JOO, Byung-Ho LEE, Dong-Seong SOHN (Future Fuel Development Division)					
Publication Place	Taejon	Publisher	KAERI	Publication Date	1999. 7.		
Page	P.50	Fig & Table	Yes(o) No()	Size	19 × 26 cm		
Note							
Classified	Open(o), Restricted(), __ Class document			Report Type	Technical report		
Sponsoring Org.				Contract No.			
Abstract (15-20 lines)		<p>Two MOX fuel rods, which are being fabricated in the Paul Scherrer Institute (PSI), Switzerland in cooperation with the Korea Atomic Energy Research Institute (KAERI), are going to be irradiated in the HBWR (Halden Boiling Water Reactor) from the beginning of 2000 in the framework of OECD Halden Reactor Programme (HRP) together with a reference MOX fuel rod supplied by the BNFL.</p> <p>Since fuel temperature, which is influenced by radial power distribution, is a basic property in analyzing fuel behavior, it is required to consider radial power distribution in the HBWR. A subroutine FACTOR_HBWR that calculates radial power density distribution for three MOX fuel rods have been developed based on neutron physics results. The newly developed subroutine FACTOR_HBWR gives good agreement with the physics calculation except slight underprediction in the central part and a little overprediction at the outer part of the pellet. The subroutine will be incorporated into a computer code COSMOS and used to analyze the in-reactor behavior of the three MOX fuel rods during the Halden irradiation test.</p>					
Subject keywords		Radial power depression, Power density factor, MOX, HBWR, HRP, KAERI, PSI, FACTOR, COSMOS					